
This report has two objectives. First, we present an original method of proof of soundness of a weakest precondition calculus, based on the notion of blocking semantics. The method mimics, at the level of logic specifications, the classical proof of type soundness. Moreover, the proof is performed formally using the Why3 environment for deductive verification, and we illustrate, along the development of the case study, the advanced features of Why3 we used. The result is a revisited presentation the weakest precondition calculus which is easy to follow, although formally made, thanks in particular to the high degree of proof automation that allows us to focus on the key points.

The Why3 system [START_REF] Bobot | Why3: Shepherd your herd of provers[END_REF] is an environment for deductive program verification. It allows the user to write programs annotated with logical specifications, and then generate proof obligations that guarantee the functional correctness.

Why3 can submit these obligations to a large collection of provers: automatic ones such as SMT solvers (Alt-Ergo [START_REF] Bobot | The Alt-Ergo automated theorem prover[END_REF], CVC3 [START_REF] Barrett | CVC3[END_REF], Z3 [START_REF] De Moura | Z3, an efficient SMT solver[END_REF], etc.) or interactive proof assistants like Coq [START_REF] Bertot | Interactive Theorem Proving and Program Development[END_REF] or PVS [START_REF]The PVS system[END_REF].

In the spectrum of deductive verification tools, Why3 positions itself in the middle of interactive environments, that offer highly expressive languages but a low level of automation (Coq, PVS, Isabelle/HOL, etc.), and more automated systems but equipped with poorer languages (Spec#, VCC, Frama-C, KeY, etc.). A natural objective is to catch the best possible between these extremes: a sufficiently expressive language that still permits significantly automated proofs.

A first objective of this report is to illustrate the capabilities of Why3 on some case study, that involves complex objects like recursive algebraic data types and inductive predicate definitions, so as to illustrate how proofs can be automatized in such a context. A class of examples involving such kind of objects is given by the symbolic computation methods, in particular the tools for analyzing programs: compilers, static analyzers, etc. We chose to formalize a calculus of weakest precondition, partially following a recent approach due to Herms et al. [START_REF] Herms | A certified multi-prover verification condition generator[END_REF] for the certification of a verification condition generator, using Coq. This new approach is based on a notion of big-step blocking semantics, defined co-inductively. The second objective of this work is to present an original approach of the weakest precondition calculus. The originality resides in the use of a small-steps blocking semantics, which is moreover formalized using the Why3 language, significantly restricted with respect to those of Coq.

Apart from the algebraic data types and the inductive predicates, the advances features of Why3 that we are going to use concern the tools available to build proofs. First, we use a transformation of Why3 to make structural inductions. Second, when the goals to prove get too complex to be proved fully automatically, we call the Coq back-end, and within Coq we use the « why3 » tactic that can finish the proof of a sub-goal by calling automated provers through Why3. These features allow us to make in Coq only the subtle parts of the proofs.

Our case study is structured into components, named theories in Why3, and this report follows the same structure, represented on Figure 1. In Section 2 we define the language under study: its syntax (Section 2.1), its operational semantics (Section 2.2) and its typing rules (Section 2.3). In Section 2.4 we formulate and prove classical results of preservation of typing by reduction. Section 3 is devoted to the notion of substitution and to the important problems related to fresh variables. Although not needed for the remaining, Section 4 presents the classical Hoare logic rules and prove their soundness. The main results are then presented in Section 5: definition of the weakest precondition calculus, and soundness results based on the small-steps blocking semantics.

Formalization of Toy Imperative Language

This section formalize our language which contains assignment, sequence, a conditional statement and a while loop.

For each of the theories we will present, we are going to tell how the proof of the lemmas are obtained, by giving the table that is automatically generated by Why3 from the proof sessions [START_REF] Bobot | The Why3 platform, version 0.80[END_REF].

In most cases, provers are launched with a time limit of 5 seconds and a memory limit of 1 gigabyte. The cases where either one of these limits is reach are indicated in the tables below by results between parentheses.

Syntax

The first theory defines our language syntax. Traditionally, one defines language syntax with grammars. Here, we directly give Why3 definitions under the form of algebraic data type declarations. We can not present the Why3 syntax in detail in this report, you may refer to the manual [START_REF] Bobot | The Why3 platform, version 0.80[END_REF] or J.-C.Filliâtre lecture given at JFLA 2012 1 . The syntax is similar to the OCaml one, we hope it is intuitive enough for the reader.

Our language has three base types which correspond to three sets of values. We introduce abstract types for identifiers, by making a distinction between identifiers for mutable variables, and identifiers for logic variables.

Formulas of the logic language include terms (if Boolean), classical connectors for conjunction, negation and implication, and universal quantification. These are the only needed connectors, but others could be added. We also add local binding with let that allows us to write more elegant rules in the weakest precondition calculus. Note that we do not use any advanced technique such as De Bruijn indexes for local bindings. Difficulties related to variable naming will be handled in Section 3. We consider assignment of a mutable variable by a term, sequences (skip constant representing an empty sequence), standard conditional statement and a while loop.

Unconventionally, we add logic annotations directly in the syntax: explicit invariants for loops, as well as assert statements.

This first theory contains only definitions and no lemmas. Therefore, there is nothing to prove.

Operational semantics

The aim of the second theory is to define the operational semantics of our language. We choose a small step version.

In a first step, we define evaluation environments: for mutable variable, on one hand, and for logic variables on the other. get _ stack function provides a shortcut to access to an element; it is defined by induction on the list. Note that this function returns void if we try to access to a variable absent from the list. This case never happen on well-typed programs as defined in the next section. Finally, we pose lemmas showing a similar behavior to maps, that are automatically proved.

lemma get _ stack _ eq: forall x:ident, v:value, pi:stack. get _ stack x (Cons (x,v) pi) = v lemma get _ stack _ neq: forall x i:ident, v:value, pi:stack.

x = i → get _ stack i (Cons (x,v) pi) = get _ stack i pi
Next, we define an auxiliary function for evaluating binary operations. This definition is made by patternmatching on the values and the operator. Since in Why3 functions must be total, we complete our definition by returning void in ill-typed cases. 2 . We can now define evaluation of terms as a total recursive function that returns a value, and evaluation of formulas by a total recursive predicate. These two functions take a current state of a program, given by two environments Σ and Π, as arguments. → eval _ fmla sigma pi f1 ∧ eval _ fmla sigma pi f2

| Fnot f → not (eval _ fmla sigma pi f) | Fimplies f1 f2 → eval _ fmla sigma pi f1 → eval _ fmla sigma pi f2 | Flet x t f → eval _ fmla sigma (Cons (x,eval _ term sigma pi t) pi) f | Fforall x TYint f → forall n:int. eval _ fmla sigma (Cons (x,Vint n) pi) f | Fforall x TYbool f → forall b:bool. eval _ fmla sigma (Cons (x,Vbool b) pi) f | Fforall x TYunit f → eval _ fmla sigma (Cons (x,Vvoid) pi) f end
Note that in the case of binders Flet and Fforall, the linked variable x is added to the stack. In the following, we will note these functions in the form t Σ,Π and f Σ,Π .

Finally, we define the notion of valid formula, in the sense that it is true in any state.

predicate valid _ fmla (p:fmla) = forall sigma:env, pi:stack. eval _ fmla sigma pi p

Execution of a statement in a given state is classically defined by a one step reduction relation noted

Σ, Π, s Σ , Π , s
Since there are statements that do not reduce, we cannot formalize this relation by a Why3 function 3 . Instead, we use an inductive predicate of Why34 . Our definition is a formalization close to the usual representation using inference rules. It is shown on Figure 2. Thus, the one _ step _ while _ true case correspond to the rule inv Σ,Π cond Σ,Π = T rue Σ, Π, while cond invariant inv do body Σ, Π, body; while cond invariant inv do body

We use a blocking semantic in the way proposed by Herms et al. [START_REF] Herms | A certified multi-prover verification condition generator[END_REF]: The execution blocks whenever an invalid assertion is met or if a loop invariant is broken when the loop condition is checked (as shown in the rule above). Conversely, assuming that the program is well-typed, the only reason why a statement would not execute is that one of its annotations is not respected.

The advantage of such a definition is that we define the fact that a program respect its specifications by the fact that it executes without blocking [START_REF] Herms | A certified multi-prover verification condition generator[END_REF], and this applies in particular for a program does not terminate.

Another inductive predicate defines the execution in n steps of a program, by a reflexive-transitive closure, shown on Figure 3. Remark that we make explicit the number n of steps of execution in this definition, to do proofs by induction on n. Notice that the built-in int type in Why3 is the type of integers (signed), there is no built-in type for natural numbers only. Thus, to make well-founded inductions, we pose the following elementary, but essential, lemma. To end this theory, we define the notion of reducibility of a program in a given state.

predicate reducible (sigma:env) (pi:stack) (s:stmt) = exists sigma':env, pi':stack, s':stmt. one _ step sigma pi s sigma' pi' s'

So, for the case of our theory modeling the operational semantics, there are 3 lemmas to prove. Results are shown on Figure 4. As expected, the lemma steps _ non _ neg cannot be proved by automated provers, as it requires an induction on the many _ steps predicate. We do the proof in Coq, and this proof is easy (one line): induction 1; auto with zarith.

Typing

This theory introduces typing of programs. At first, we define a total function that returns a type of some value.

function type _ value (v:value) : datatype = match v with | Vvoid → TYunit | Vint _ → TYint | Vbool _ → TYbool end
Type of the binary operators of our language is given by a simple inductive predicate. 

Relations between typing and operational semantic

This theory contains our first results linking typing and execution. At first, it defines a predicate for compatibility between typing and evaluation environments, that is, to each identifier of type t is associated a value of type t.

predicate compatible _ env (sigma:env) (sigmat:type _ env) (pi:stack) (pit: type _ stack) = (forall id: mident. type _ value (get _ env id sigma) = get _ reftype id sigmat) ∧ (forall id: ident. type _ value (get _ stack id pi) = get _ vartype id pit)

Then we pose an inversion lemma, that allows us to know which is the constructor of a value, from its type.

lemma type _ inversion : forall v:value. match (type _ value v) with | TYbool → exists b: bool. v = Vbool b | TYint → exists n: int. v = Vint n | TYunit → v = Vvoid end
The results of proofs of all the lemmas in this theory are given on Figure 8. For this particular lemma, an obvious case-based reasoning allows to prove such lemma. Nevertheless, theorem provers are not able to prove this goal 5 . We could do this proof with Coq, but we can also use Why3 features: goals transformations. In Why, there exists such a transformation to make structural induction. Here, the type value is not recursive, so it amounts to a case-based reasoning. The three generated sub-goals are automatically discharged. We also made a Coq proof of type _ inversion lemma. It can be done easily thanks to another feature of Why3: the why3 tactic of Coq. It is a tactic loaded as Coq plug-in by Require Import Why3. Ltac ae := why3 "alt-ergo" timelimit 5

Proof obligations

Alt-Ergo (0.93. The tactic ae is therefore a shortcut to call Alt-Ergo from Coq, with a timeout fixed to 5 seconds. The tactic allows, then, to call (through Why3) theorem provers, according to the architecture of Figure 9.

The proof of type _ inversion lemma is then

destruct v; ae.
It should be noted that this tactic does not produce any proof term, so we must trust both the way it reconstructs a Why3 task from the current Coq goal, and theorem provers called after, the same way we trust these theorem provers when called directly from Why3.

The first significant result states that a well-typed term of type t, is evaluated to a value of type t, too. This lemma has to be proved by structural induction on t. Once again, instead of making proof entirely in Coq, we first apply Why3 transformation for structural induction, which produces 4 new sub-goals, corresponding to a standard induction schema. Among these sub-goals, 3 are automatically proved (See Figure 8), the only one that we have to prove with Coq concerns binary operations which is done in 7 lines: In total, this proof of preservation of typing by reduction is only three lines long in Coq! Notice that 5 of the 7 seconds (see Figure 8) are lost when waiting for Alt-Ergo failing on the 8th sub-goal.

Substitutions, fresh variables

For defining our weakest precondition calculus, we will need to substitute variables. We dedicate a specific theory to the operation of substitution and its properties, that are particularly expressed under hypothesis of freshness of variables. Naming and freshness problems are classically difficult points to handle when we formalize languages with binders, as shown by the famous POPLmark challenge [START_REF] Aydemir | Mechanized metatheory for the masses: The POPLmark challenge[END_REF]. We propose here a basic approach, that we believe is easy to deal with, and sufficient for our study. First, we define the substitution operation, where we do not care about the problem of variable capture. The only one operation that we are interested in is to substitute a mutable program variable by a logical variable.

( ** substitution of a mutable variable [x] by a logic variable [v] warning: proper behavior only guaranteed if [v] 

→ Fand (msubst f1 x v) (msubst f2 x v) | Fnot f → Fnot (msubst f x v) | Fimplies f1 f2 → Fimplies (msubst f1 x v) (msubst f2 x v) | Flet y t f → Flet y (msubst _ term t x v) (msubst f x v) | Fforall y ty f → Fforall y ty (msubst f x v) end RR n°8185
Then, we define the notion of fresh variable, that is, does not appear in a term or a formula. These are natural recursive definitions.

( ** [fresh _ in _ term id t] is true when [id] does not occur in [t] * ) predicate fresh _ in _ term (id:ident) (t:term) = match t with

| Tvalue _ | Tderef _ → true | Tvar i → id = i | Tbin t1 _ t2 → fresh _ in _ term id t1 ∧ fresh _ in _ term id t2 end predicate fresh _ in _ fmla (id:ident) (f:fmla) = match f with | Fterm e → fresh _ in _ term id e | Fand f1 f2 | Fimplies f1 f2 → fresh _ in _ fmla id f1 ∧ fresh _ in _ fmla id f2 | Fnot f → fresh _ in _ fmla id f | Flet y t f → id = y ∧ fresh _ in _ term id t ∧ fresh _ in _ fmla id f | Fforall y ty f → id = y ∧ fresh _ in _ fmla id f end
For reasoning about substitutions in the remaining of our formalization, we need some general lemmas we present now. Unsurprisingly, these lemmas have freshness hypothesis on concerned variables.

The two first lemmas relate the evaluation of a term (resp. a formula) on which we apply a substitution, with the evaluation of this term (resp. formula) in a modified state. In other words we have the identity These lemmas are proved by structural induction on the term (resp. formula), thanks to the Why3 transformation.

t[x ← v] Σ,Π = t Σ[x←Π(v)],
All sub-goals are automatically discharged (see Figure 10), except two which must be proved in Coq (case Flet and Fforall). These proofs are simple:

destruct f; auto. simpl; ae.

The next lemmas allow us, when evaluating a term or a formula, to swap two consecutive identifiers in the stack

(if they are different), that is t Σ,Π1•(id1,v1)•(id2,v2)•Π2 = t Σ,Π1•(id2,v2)•(id1,v1)•Π2
. The first of these lemmas, for terms, is written as6 lemma eval _ swap _ term: forall t:term, sigma:env, pi l:stack, id1 id2:ident, v1 v2:value. id1 = id2 → eval _ term sigma (l++(Cons (id1,v1) (Cons (id2,v2) pi))) t = eval _ term sigma (l++(Cons (id2,v2) (Cons (id1,v1) pi))) t

and is, again, proved by structural induction (See Figure 11). There is only one sub-goal which is not automatically discharged, the one corresponding to the case Tvar. This case is actually subtle, because it requires another induction, on the list l. In Coq, we use induction l tactic, and we finish the 2 sub-goals with why3 tactic.

Proof obligations

Alt-Ergo (0.93. We also pose the following instance of eval _ swap _ gen for the case l=Nil.

lemma eval _ swap: forall f:fmla, sigma:env, pi:stack, id1 id2:ident, v1 v2:value.

id1 = id2 → (eval _ fmla sigma (Cons (id1,v1) (Cons (id2,v2) pi)) f ↔ eval _ fmla sigma (Cons (id2,v2) (Cons (id1,v1) pi)) f)
While it is sufficient to instantiate the preceding lemma, this lemma is not proved by any theorem prover (see Figure 13). It has to be done in Coq intros f sigma pi id1 id2 v1 v2 h1. apply eval _ swap _ gen with (l:=Nil); auto.

Finally, two last lemmas allow to pop from the stack an identifier that does not appear in the evaluated term (resp. formula), that is t Σ,(id,v)•Π = t Σ,Π if id is fresh. lemma eval _ term _ change _ free : forall t:term, sigma:env, pi:stack, id:ident, v:value. fresh _ in _ term id t → eval _ term sigma (Cons (id,v) pi) t = eval _ term sigma pi t lemma eval _ change _ free : forall f:fmla, sigma:env, pi:stack, id:ident, v:value. fresh _ in _ fmla id f → (eval _ fmla sigma (Cons (id,v) pi) f ↔ eval _ fmla sigma pi f) Once again, the proof is done by structural induction (See Figure 14). The only sub-case which is not automatically proved is the case of Fforall in the direct way of implication, that is proved with: destruct f; auto. simpl; intros H sigma pi id v (h1 & h2). destruct d; intros; rewrite <-(H _ _ id v); ae.

Hoare Logic

This section is in fact not needed for the remaining of the case study. We just illustrate how one can formalize the standard Hoare logic rules. The notion of partial correctness of a Hoare triple, usually denoted {p}s{q}, is defined as follows.

predicate valid _ triple (p:fmla) (s:stmt) (q:fmla) = forall sigma:env, pi:stack. eval _ fmla sigma pi p → forall sigma':env, pi':stack, n:int. many _ steps sigma pi s sigma' pi' Sskip n → eval _ fmla sigma' pi' q

RR n°8185

Proof obligations

Alt-Ergo (0.93. 

Figure 14: Substitution and freshness First, the consequence rule below can be proved automatically.

lemma consequence _ rule: forall p p' q q':fmla, s:stmt.

valid _ fmla (Fimplies p' p) → valid _ triple p s q → valid _ fmla (Fimplies q q') → valid _ triple p' s q'

RR n°8185

The rule for sequence is also proved automatically, but only after posing the standard "sequence lemma": lemma many _ steps _ seq: forall sigma1 sigma3:env, pi1 pi3:stack, s1 s2:stmt, n:int. many _ steps sigma1 pi1 (Sseq s1 s2) sigma3 pi3 Sskip n → exists sigma2:env, pi2:stack, n1 n2:int. many _ steps sigma1 pi1 s1 sigma2 pi2 Sskip n1 ∧ many _ steps sigma2 pi2 s2 sigma3 pi3 Sskip n2 ∧ n = 1 + n1 + n2 lemma seq _ rule: forall p q r:fmla, s1 s2:stmt.

valid _ triple p s1 r ∧ valid _ triple r s2 q → valid _ triple p (Sseq s1 s2) q

The proof of that lemma, even on paper, is not completely trivial and deserves a few lines, with an induction on the number n of steps. The proof is thus done in Coq, and is not surprising: In all these five cases, Coq must be used, and are done quite smoothly by inversion of predicates many _ steps and one _ step, and finishing the sub-goals with the Why3 tactic.

Theorem skip _ rule : forall (q:fmla), (valid _ triple q Sskip q). intros q sigma pi. intros H sigma' pi' n H1. inversion H1; subst; clear H1; auto. inversion H0. Qed.

Theorem assign _ rule : forall (p:fmla) (x:mident) (id:ident) (t:term), (fresh _ in _ fmla id p) → (valid _ triple (Flet id t (msubst p x id))

(Sassign x t) p). intros p x id t h1. red; intros. inversion H0; subst; clear H0. inversion H1; subst; clear H1. inversion H2; subst; clear H2. ae. ae. Qed.

Theorem if _ rule : forall (t:term) (p:fmla) (q:fmla) (s1:stmt) (s2:stmt), ((valid _ triple (Fand p (Fterm t)) s1 q) ∧ (valid _ triple (Fand p (Fnot (Fterm t))) s2 q)) → (valid _ triple p (Sif t s1 s2) q). unfold valid _ triple. intros t p q s1 s2 (h1,h2 

→ q | Sassert f → Fand f (Fimplies f q)
( ** asymmetric and * ) | Sseq s1 s2 → wp s1 (wp s2 q) | Sassign x t → let id = fresh _ from q in Flet id t (msubst q x id)

| Sif t s1 s2 → ( * (t → WP(s1,Q)) ∧ (not t → WP(s2,Q)) * ) Fand (Fimplies (Fterm t) (wp s1 q)) (Fimplies (Fnot (Fterm t)) (wp s2 q)) | Swhile cond inv body → ( * inv ∧ forall effects, (cond ∧ inv → WP(body,inv)) ∧ (not cond ∧ inv → Q) * )
Fand inv (abstract _ effects body (Fand (Fimplies (Fand (Fterm cond) inv) (wp body inv)) (Fimplies (Fand (Fnot (Fterm cond)) inv) q))) end Our definition above follows the classical schema for such a calculus. For the case of Sassert f, we use f ∧ (f → Q) instead of f ∧ Q as it is done in practice 7 . For an assignment Sassign x t, we substitute the mutable variable x by a fresh logic variable id, computed thanks to a auxiliary function fresh _ from, then we bind id to the value assigned t. The case of the loop classically introduces a quantification on the side effects of the body of that loop. This quantification is realized by calling another auxiliary function abstract _ effects.

In order to deal with difficulties separately, we do not detail implementations of those auxiliary functions, we only axiomatize them. The first function is axiomatized by: function fresh _ from (f:fmla) : ident axiom fresh _ from _ fmla: forall f:fmla. fresh _ in _ fmla (fresh _ from f) f

In other words, for any formula f , fresh _ from f returns an arbitrary identifier that is fresh in f ; we don't care about how it is computed.

The function that abstract the side effects has type function abstract _ effects (s:stmt) (f:fmla) : fmla and its expected behavior is, for given statement s and formula P , to return the formula ∀w, f where w = w 1 , . . . , w k is the set of assigned variables in s. Our axiomatization of this behavior is made of four hypotheses detailed below. The first hypothesis formalizes the specialization property: if Σ, Π |= ∀w, f then Σ, Π |= f . axiom abstract _ effects _ specialize : forall sigma:env, pi:stack, s:stmt, f:fmla.

eval _ fmla sigma pi (abstract _ effects s f) → eval _ fmla sigma pi f

The second hypothesis formalizes the distributivity of quantification over conjunction : if Σ, Π |= (∀w, P )∧(∀w, Q) then Σ, Π |= ∀w, P ∧ Q.

axiom abstract _ effects _ distrib _ conj : forall s:stmt, p q:fmla, sigma:env, pi:stack.

eval _ fmla sigma pi (abstract _ effects s p) ∧ eval _ fmla sigma pi (abstract _ effects s q) → eval _ fmla sigma pi (abstract _ effects s (Fand p q))

The third hypothesis is a bit more subtle: it expresses a property of the quantification on the two parts of an implication.

axiom abstract _ effects _ monotonic : forall s:stmt, p q:fmla. valid _ fmla (Fimplies p q) → forall sigma:env, pi:stack. eval _ fmla sigma pi (abstract _ effects s p) → eval _ fmla sigma pi (abstract _ effects s q)

In other words, if |= P → Q then |= (∀w, P ) → (∀w, Q). This property can be seen as a consequence of two other properties: first, if |= f then |= ∀w, f ; second, if ∀w, (P → Q) then (∀w, P ) → (∀w, Q). It is important to notice that this hypothesis talks about validity in all states, because for fixed Σ and Π, Σ, Π |= P → Q does not imply Σ, Π |= (∀w, P ) → (∀w, Q). A counter-example is as follows: assuming Σ such that Σ(x) = 42, then true → x = 42 is valid, but not (∀x, true) → (∀x, x = 42).

Until now, the hypotheses on abstract _ effects express abstractly that it behaves as an universal quantification, without saying precisely on what. The fourth and last hypothesis we pose now expresses on what variables we quantify: if w = w 1 , . . . , w k is the set of variables assigned by s, then ∀w, P is a formula that is invariant by the WP calculus through s. axiom abstract _ effects _ writes : forall sigma:env, pi:stack, s:stmt, q:fmla. eval _ fmla sigma pi (abstract _ effects s q) → eval _ fmla sigma pi (wp s (abstract _ effects s q))

In fact, this abstract way of specifying the behavior abstract _ effects is precisely the important property that one tries to obtain when introducing the universal quantification into the weakest precondition of a loop: the formula cond ∧ inv → WP(body, inv) must be quantified, so that it becomes true (because it is invariant) at each iteration of the loop.

Basic Properties of the Calculus

In the literature, there exist advanced theoretical studies on the family of predicate transformers in general, that includes the weakest precondition calculus [START_REF] Back | Refinement calculus -a systematic introduction[END_REF]. Several properties are stated, often without showing why they are useful. In our case study, we are going to state and prove two properties that are needed for the main proof of soundness.

The first of these two properties is classically named monotonicity of the calculus. It is stated mathematically under the form: for any statement s and any formulas P and Q, if |= P → Q then |= WP(s, P ) → WP(s, Q). In Why3 it becomes:

Proof obligations

Alt-Ergo (0.93. valid _ fmla (Fimplies p q) → valid _ fmla (Fimplies (wp s p) (wp s q))

Notice that, as before, quantification on all states is essential: Σ, Π |= P → Q does not necessarily imply Σ, Π |= WP(s, P ) → WP(s, Q). There is a similar counter-example: if Σ(x) = 42 then (true → x = 42) but WP(x := 7, true) = true does not imply WP(x := 7, x = 42) = (7 = 42).

The proof of the monotonicity property is done by induction on the structure of s. In Why3, after application of the transformation for structural induction, it gives 6 sub-goals, as shown on Figure 16.

Only the cases of the assignment, the conditional and the loop are not proved automatically. The proofs for the case of Sassigns and Sif can both be done using the 2 same lines: destruct s; auto. unfold valid _ fmla; simpl; ae.

it should be noted that that last call to the why3 tactic makes use of lemmas for the substitution and fresh variables. The case of Swhile est slightly more involved, one needs to apply the hypothesis (3): destruct s; auto. unfold valid _ fmla; simpl. intros H1 p q H2; intuition. apply abstract _ effects _ monotonic with (2:=H3). unfold valid _ fmla; simpl. intuition.

The second property that we need is the distributivity over the conjunction: if Σ, Π |= WP(s, P ) and Σ, Π |= WP(s, Q) then Σ, Π |= WP(s, P ∧ Q). lemma distrib _ conj: forall s:stmt, sigma:env, pi:stack, p q:fmla. eval _ fmla sigma pi (wp s p) ∧ eval _ fmla sigma pi (wp s q) → eval _ fmla sigma pi (wp s (Fand p q))

The reciprocal is also true, but not needed. The proof proceeds again by structural induction on s, as shown on Figure 17. The cases of assignment, sequence, and loop are not proved automatically, the proofs are terminated in Coq.

The proof for the case of assignment is again easy: The proof in the case of the sequence is more complex. We need to use the preceding property of monotonicity. It indeed corresponds to an "intelligent" part of the proof, where we state as an intermediate lemma that

|= WP(s 1 , WP(s 2 , P ) ∧ WP(s 2 , Q)) → WP(s 1 , WP(s 2 , P ∧ Q)).
destruct s; auto. simpl; intros H1 H2 sigma pi p q (H3 & H4). assert (H: valid _ fmla (Fimplies (Fand (wp s2 p) (wp s2 q)) (wp s2 (Fand p q)))). unfold valid _ fmla ; simpl; ae. generalize (monotonicity s1 _ _ H). unfold valid _ fmla; simpl; ae.

In the case of the loop, we use hypotheses (3) and (4). destruct s; auto. simpl. intros H sigma pi p q ((h1 & h2) & (h3 & h4)). split; auto. apply abstract _ effects _ monotonic with (p:= (Fand (Fand (Fimplies (Fand (Fterm t) f) (wp s f))

(Fimplies (Fand (Fnot (Fterm t)) f) p)) (Fand (Fimplies (Fand (Fterm t) f) (wp s f))

(Fimplies (Fand (Fnot (Fterm t)) f) q)))). unfold valid _ fmla; simpl. intuition. apply abstract _ effects _ distrib _ conj; auto.

Soundness of the Weakest Precondition Calculus

We follow the classical approach for proving type soundness. It amounts to state two lemmas: on the one hand, the preservation of WP by reduction ; on the other hand a progress property, that is the validity of WP(s, Q) entails the reductibility of s (if s = Sskip). lemma wp _ preserved _ by _ reduction: forall sigma sigma':env, pi pi':stack, s s':stmt. one _ step sigma pi s sigma' pi' s' → forall q:fmla. eval _ fmla sigma pi (wp s q) → eval _ fmla sigma' pi' (wp s' q) RR n°8185

Proof obligations

Coq (8.3pl4) lemma wp _ preserved _ by _ reduction 6.78 (5s) (5s) (5s) (5s) 0.75 (5s) (5s)

Figure 19: Progress lemma This first lemma is proved by induction on hypothesis (one _ step sigma pi s sigma' pi' s'). This must be done in Coq (Figure 18). After using the induction tactic in Coq, there are 8 sub-goals, among which 6 can be proved simply by the simpl tactic followed by the why3 tactic. The two remaining cases are for the reduction of loops. To prove them, we start by using the hypothesis (1) to make explicit the fact that the formula (cond ∧ inv → WP(body, inv)) ∧ (¬cond ∧ inv → Q) is true in the current state, then we conclude using the why3 tactic. One should notice that these calls to why3 tactic use the hypothesis (4) and the property distrib _ conj. The progress lemma is stated as follows.

lemma progress: forall s:stmt, sigma:env, pi:stack, sigmat: type _ env, pit: type _ stack, q:fmla.

compatible _ env sigma sigmat pi pit ∧ type _ stmt sigmat pit s ∧ eval _ fmla sigma pi (wp s q) ∧ s = Sskip → reducible sigma pi s

It may seem strange that the lemma above depends on an arbitrary formula Q, that does not play any role in the conclusion. Thus, it may seem enough to state this lemma for Q = true. But then it would not be general enough to be proved by induction. For example, to prove the progress of a sequence s 1 ; s 2 knowing that WP(s 1 ; s 2 , true) holds, we need to prove the progress of s 1 knowing that WP(s 1 , WP(s 2 , true)) holds, and the formula WP(s 2 , true) is a priori arbitrary. The progress lemma is proved by structural induction on s. Only the case of Sskip is then proved automatically (Figure 19), the reason being that the conclusion reducible is existentially quantified. In each case we terminate the proof in Coq, in a few lines, with the exists tactic. For the case of the loop, we use once more the property Our main result is now: for any program s, any state Σ, Π and any formula Q, if Σ, Π |= WP(s, Q) then either s executes infinitely, or it reduces to Sskip, and in the second case Q is true in the final state. Since it is not handy to formulate that an execution is infinite, we state that slightly differently: if s executes into some finite number of steps into s such that s does not reduce anymore, then s = Sskip. All this must be stated only for well-typed programs, thus the Why3 formulation of our main result is: lemma wp _ soundness: forall n :int, sigma sigma':env, pi pi':stack, s s':stmt, sigmat: type _ env, pit: type _ stack, q:fmla. compatible _ env sigma sigmat pi pit ∧ type _ stmt sigmat pit s ∧ many _ steps sigma pi s sigma' pi' s' n ∧ not (reducible sigma' pi' s') ∧ eval _ fmla sigma pi (wp s q) → s' = Sskip ∧ eval _ fmla sigma' pi' q This is proved by induction on n, what we do calling Coq (Figure 20). For n = 0, the progress lemma ensures that s can only be Sskip. For the case n > 0, we know that s reduces in one step into some s 0 , the lemma of preservation of WP by reduction allows us to apply the induction hypothesis on s 0 .

intros n sigma sigma' pi pi' s s' sigmat pit q. intros Hcomp Htype (h1,(h2,h3)). generalize (steps _ non _ neg _ _ _ _ _ _ _ h1). intros Hnpos. generalize sigma sigma' pi pi' s s' sigmat pit q Hcomp Htype h1 h2 h3. clear sigma sigma' pi pi' s s' sigmat pit q Hcomp Htype h1 h2 

General remarks on proofs

Let's put aside the section on Hoare logic which was presented for illustration but not needed for the main results. Overall, the proofs that we needed to do in Coq amount to 142 lines of tactics. This number should be compared with the number of lines of specification that we wrote: 390, that is more than twice the number of lines of proofs. This is a really good level of automation, that invalidates the usual belief that a proof of complex behavior of a program requires significantly more lines of proofs than lines of code.

The source of this case study, together with the Why3 proof session and the Coq proofs are available on the web page http://toccata.lri.fr/gallery/WP _ revisited.en.html. A ZIP archive is provided, that contains everything needed to replay the proofs using Why3 0.80 [START_REF] Bobot | The Why3 platform, version 0.80[END_REF].

Conclusions

We have presented a technique for proving the soundness of a weakest preconditions calculus, that imitates the classical technique of proof of type soundness of a programming language, by showing both that the WP is preserved by reduction and that the validity of the WP entails the reducibility. Since the conformance of the annotations of a program is by definition the absence of blocking (blocking semantics), we obtain a verification method, guaranteed correct, even on non-terminating programs. The definition via blocking semantics is proposed by Herms et al. [START_REF] Herms | A certified multi-prover verification condition generator[END_REF] in the case of a great-step semantics, soundness being proved by co-induction. We use a small-step blocking semantics : our approach is thus original as far as we know. Indeed, a similar approach was used by Conchon and Filliâtre [START_REF] Conchon | Semi-Persistent Data Structures[END_REF] in a different context: proof of soundness of a decision procedure for the safety of use of semi-persistent data structures.

The proof obtained by our approach is not difficult, in fact we pretend this is a natural approach, simple to understand and moreover is done with highly automated proof. We plan to use this method for teaching the basics of deductive program verification 8 .

Along this case study, we identified several interesting features of the Why3 environment: structuring in theories, algebraic data types, inductive predicate, direct calls to automated provers, transformation for proving by structural induction, call to Coq to terminate difficult proofs, use of the why3 tactic to terminate the Coq proofs. At the end, the number of lines of manual proof is small: 142 lines of Coq proofs for 390 lines of program code, that is a very satisfactory degree of automation. Nevertheless, one should be aware that during the development of this study, we had to do, temporarily, more Coq proofs, in order to identify the appropriate lemmas needed to finally get good automation. The why3 tactic is by the way very useful during such a phase of development of proofs.

We also identified a few weaknesses of Why3 that could make interesting motivations for extensions. The automated provers appear to be weak as soon as a hypothesis involves an inductive predicate: a transformation to make an inversion of such a hypothesis, so as to reason on it by induction, could be very handy and useful. For example, the lemma steps _ non _ neg could be proved without the need of Coq. Similarly, a transformation for proving a lemma by induction on integers would be useful.

We noticed a general weakness of SMT provers: often they can't prove a goal that can be proved in a few lines of Coq. For example, these provers are not able to do a simple thing like eexists; eauto of Coq. We also tested some provers specialized for first-order reasoning (of the kind « TPTP » : Vampire [START_REF] Riazanov | Vampire[END_REF], Spass [START_REF] Weidenbach | Spass version 3.5[END_REF], Eprover [START_REF] Schulz | System description: E 0.81[END_REF]) without success. We also noticed that the automated provers do not handle well the goals that can be proved very easily with the simpl tactic of Coq followed by a call to the why3 tactic. This emphasizes a weakness in the support for computations within proofs by such provers.

Another kind of missing feature is the ability to pose statements about the functions of the Why3 programming language. In fact, in this case study we only defined pure functions. If a function like wp was written using the programming language (that allow side-effects), then it would not be possible to state properties of its behavior like we did, but only by giving a post-condition, and in particular it would not be possible to state a property like wp _ soundness.

The case study itself deserves to be extended. First, the functions fresh _ from and abstract _ effects should be realized. A significant extension would to handle a language with several sub-programs, like it was done by Herms et al. [START_REF] Herms | A certified multi-prover verification condition generator[END_REF] and was indeed a reason to use the blocking semantics approach.

In a longer term, one could seriously think about using this approach to develop code correct by construction. In fact, an extension of Why3 currently in progress is the ability to extract OCaml code. With respect to an approach fully in Coq, we would gain a lot of automation.

 1 Figure 1: Structure of the development

  type fmla = ( ** Formulas * ) | Fterm term ( ** terms: atomic formulas * ) | Fand fmla fmla ( ** conjunction * ) | Fnot fmla ( ** negation * ) | Fimplies fmla fmla ( ** implication * ) | Flet ident term fmla ( ** local binding: let id = term in fmla * ) | Fforall ident datatype fmla ( ** universal quantification: forall id : ty, fmla * ) Finally, the syntax of statements of our imperative language follows. type stmt = ( ** Statements * ) | Sskip ( ** no-op statement * ) | Sassign mident term ( ** assignment id := term * ) | Sseq stmt stmt ( ** sequence * ) | Sif term stmt stmt ( ** conditional statement * ) | Sassert fmla ( ** assertion statement * ) | Swhile term fmla stmt ( ** while loop with condition, invariant and body * )

Figure 2 :Figure 3 :

 23 Figure 2: One-step reduction

  inductive type _ operator (op:operator) (ty1 ty2 ty: datatype) = | Type _ plus : type _ operator Oplus TYint TYint TYint | Type _ minus : type _ operator Ominus TYint TYint TYint | Type _ mult : type _ operator Omult TYint TYint TYint | Type _ le : type _ operator Ole TYint TYint TYbool We define typing environments similarly to evaluation ones. type type _ stack = list (ident, datatype) ( ** map local immutable variables to their type * ) function get _ vartype (i:ident) (pi:type _ stack) : datatype = match pi with | Nil → TYunit | Cons (x,ty) r → if x=i then ty else get _ vartype i r end type type _ env = IdMap.map mident datatype ( ** map global mutable variables to their type * ) function get _ reftype (i:mident) (e:type _ env) : datatype = IdMap.get e i Typing judgments for terms, formulas and statements are, then, naturally defined by new inductive predicates, given on Figures 5, 6 and 7 inductive type _ term type _ env type _ stack term datatype = | Type _ value : forall sigma: type _ env, pi:type _ stack, v:value. type _ term sigma pi (Tvalue v) (type _ value v) | Type _ var : forall sigma: type _ env, pi:type _ stack, v: ident, ty:datatype. (get _ vartype v pi = ty) → type _ term sigma pi (Tvar v) ty | Type _ deref : forall sigma: type _ env, pi:type _ stack, v: mident, ty:datatype. (get _ reftype v sigma = ty) → type _ term sigma pi (Tderef v) ty | Type _ bin : forall sigma: type _ env, pi:type _ stack, t1 t2 : term, op:operator, ty1 ty2 ty:datatype.type _ term sigma pi t1 ty1 → type _ term sigma pi t2 ty2 → type _ operator op ty1 ty2 ty → type _ term sigma pi (Tbin t1 op t2) ty

Figure 5 :Figure 6 :Figure 7 :

 567 Figure 5: Typing rules for terms

  is fresh * ) function msubst _ term (t:term) (x:mident) (v:ident) : term = match t with | Tvalue _ | Tvar _ → t | Tderef y → if x = y then Tvar v else t | Tbin t1 op t2 → Tbin (msubst _ term t1 x v) op (msubst _ term t2 x v) end function msubst (f:fmla) (x:mident) (v:ident) : fmla = match f with | Fterm e → Fterm (msubst _ term e x v) | Fand f1 f2

Π

  lemma eval _ msubst _ term: forall e:term, sigma:env, pi:stack, x:mident, v:ident. fresh _ in _ term v e → eval _ term sigma pi (msubst _ term e x v) = eval _ term (IdMap.set sigma x (get _ stack v pi)) pi e lemma eval _ msubst: forall f:fmla, sigma:env, pi:stack, x:mident, v:ident. fresh _ in _ fmla v f → (eval _ fmla sigma pi (msubst f x v) ↔ eval _ fmla (IdMap.set sigma x (get _ stack v pi)) pi f)

Figure 18 :

 18 Figure 18: Preservation by reduction Proof obligations Alt-Ergo (0.93.1) Alt-Ergo (0.94) CVC3 (2.2) CVC3 (2.4.1) Coq (8.3pl4) Z3 (3.2) Z3 (4.2)

  Type env denotes environments for mutable variables. It is a map that associates a value to any mident. For this we use the Why3 standard library generic type map.

use export Syntax use map.Map as IdMap type env = IdMap.map mident value ( ** map global mutable variables to their value * ) function get _ env (i:mident) (sigma:env) : value = IdMap.get sigma i get _ env function provides a shortcut for accessing to the elements of an environment. Type stack denotes environments for logic variables: a stack of pairs (ident, value). We use the generic type list of Why3 standard library. use export list.List type stack = list (ident, value) ( ** map local immutable variables to their value * ) function get _ stack (i:ident) (pi:stack) : value = match pi with | Nil → Vvoid | Cons (x,v) r → if x=i then v else get _ stack i r end

  This lemma is proved in Coq, by induction on one _ step hypothesis. We obtain 8 sub-cases, 7 of them are automatically discharged by why3 tactic. The last sub-goal, the sequence, requires the hypothesis of well-typedness of the sequence before calling why3 tactic.

	destruct t; auto.
	simpl; intros.
	inversion H2; subst; clear H2.
	destruct H9 as (h1 & h2 & h3).
	generalize (type _ inversion (eval _ term sigma pi t1)).
	generalize (type _ inversion (eval _ term sigma pi t2)).
	destruct h3; ae.
	The second result is preservation of well-typing by reduction.
	lemma type _ preservation : forall s1 s2:stmt, sigma1 sigma2:env, pi1 pi2:stack,
	sigmat:type _ env, pit:type _ stack.
	type _ stmt sigmat pit s1 ∧ compatible _ env sigma1 sigmat pi1 pit ∧
	one _ step sigma1 pi1 s1 sigma2 pi2 s2 →
	type _ stmt sigmat pit s2 ∧ compatible _ env sigma2 sigmat pi2 pit
	intros s1 s2 sigma1 sigma2 pi1 pi2 sigmat pit (h1,(h2,h3)).
	induction h3; try ae.
	inversion h1; subst; clear h1; ae.

http://why3.lri.fr/jfla-2012/ RR n°8185

Note that True and False are the constructors of Boolean type of Why3.RR n°8185

We remind that Why3 functions are always total.

The notion of inductive predicate in Why3 is similar to Coq or PVS ones: It is the smallest relation verifying the given clauses. Why3 verifies the usual monotonicity conditions for such definitions.RR n°8185

Indeed, we realized later in the development that CVC3 2.2 proves that lemma RR n°8185

Symbol ++ denotes concatenation of lists.RR n°8185

It forces f to appear as hypothesis in the other proof obligations.RR n°8185

